In[1]:=
```
"---------------------------------------------------------------------";
"INSTABILITA' DI ASTA DEFORMABILE PRESSOINFLESSA
 File Mathematica

 Corso di Fondamenti di Dinamica e Instabilita' delle Strutture
 Universita' di Bergamo, Facolta' di Ingegneria, Dalmine
 prof. Egidio Rizzi
 Giugno 2007";
"---------------------------------------------------------------------";

"Analisi di stabilita' di asta flessionalmente deformabile pressoinflessa,
 incastrata al piede, soggetta a carico distribuito q, forza F e coppia W
 in sommita' (positive se tendono le medesime fibre tese, ad es. a ds.
 dell'asta) e azione assiale di compressione P. Ascissa x dall'incastro
 al piede, f freccia in sommita'.
 Approccio risolutivo con metodo statico mediante scrittura dell'equazione
 della linea elastica al II ordine con effetti di non-linearita' geometrica
 (scrittura del momento rispetto alla configurazione deformata)";

"Istruzioni d'uso:
 Ogni cella di comandi puo' essere eseguita in Mathematica
 cliccando col mouse nello spazio all'interno dei delimitatori
 visibili a destra e agendo sulla tastiera con sfhit+enter";

"Disabilita la segnalazione di spelling errors";
Off[General::spell]
Off[General::spell1]

"Equazione del momento flettente legato ai soli carichi trasversali";
Mc = W + F (l - x) + q (l - x) ^2 / 2;

"Soluzione standard della linea elastica senza effetti del II ordine
 dovuti a P";
ye = Integrate[Integrate[Mc / EJ, x], x] + a1 / EJ x + a2 / EJ;
yep = D[ye, x];

"Imposizione delle condizioni al contorno in corrispondenza
 dell'incastro al piede";
solccye = Solve[{{ye /. x → 0} == 0 ,
                {yep /. x → 0} == 0}, {a1, a2}][[1]];
yefin = Simplify[ye /. solccye]

"Freccia elastica fe in sommita' dovuta a q, F, W";
fe = Simplify[yefin /. {x → l}]

"Rappresentazione della linea elastica per W=Fl, q=F/l
 a meno del fattore fe";
yefinplot = Simplify[yefin / fe /. {W → F l, q → F / l, x → xa l}];
Plot[yefinplot, {xa, 0, 1},
     PlotRange → {{0, 1}, {0, 1}}, AxesLabel → {x / l, "ye/fe"},
     AxesOrigin → {0, 0},
     PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]}}]
```

Out[16]= $\frac{x^2\,(12\,F\,l + 6\,l^2\,q + 12\,W - 4\,F\,x - 4\,l\,q\,x + q\,x^2)}{24\,EJ}$

Out[18]= $\frac{l^2\,(8\,F\,l+3\,l^2\,q+12\,W)}{24\,EJ}$



Out[21]= - Graphics -

In[22]:=
```
"Equazione della linea elastica con effetti del II ordine:
 y''+ alpha^2 y = Mc(x)/EJ + alpha^2 f, con alpha^2=P/EJ";

"Membro di destra";
md = Mc / EJ + alpha^2 f;

"Integrale particolare quadratico in x";
ypart = b + b1 x + b2 x^2 / 2;
ypartp = D[ypart, x];
ypartpp = D[ypartp, x];

diff = Simplify[ypartpp + alpha^2 ypart - md];
Collect[diff, x];

solb = Simplify[
       Solve[{{diff /. {x -> 0}} == 0,
              Coefficient[diff, x] == 0,
              Coefficient[diff, x^2] == 0}, {b, b1, b2}]];

ypartfin = ypart /. solb[[1]]

"Controllo che l'integrale particolare trovato sia effettiva soluzione";
Simplify[D[D[ypartfin, x], x] + alpha^2 ypartfin - md]
```

Out[32]= $\frac{2\,\mathrm{alpha}^4\,EJ\,f-2\,q+\mathrm{alpha}^2\,(2\,F\,l+l^2\,q+2\,W)}{2\,\mathrm{alpha}^4\,EJ}-\frac{(F+l\,q)\,x}{\mathrm{alpha}^2\,EJ}+\frac{q\,x^2}{2\,\mathrm{alpha}^2\,EJ}$

Out[34]= 0

```
In[35]:= "Integrale totale dell'equazione della linea elastica";
         ygenoa = A Sin[alpha x] + B Cos[alpha x];
         y = ygenoa + ypartfin;
         yp = D[y, x];
         ypp = D[yp, x];

         "Controllo che l'integrale totale trovato sia vera soluzione";
         Simplify[ypp + alpha^2 y - md]

Out[41]= 0

In[42]:= "Imposizione delle condizioni al contorno in corrispondenza
          dell'incastro al piede";
         solccy = Solve[{{y /. x → 0} == 0 ,
                         {yp /. x → 0} == 0,
                         {y /. x → l} - f == 0}, {A, B, f}][[1]];

         "Linea elastica finale e linea elastica finale a meno del fattore fe";
         yfin = Simplify[y /. solccy]
         yfinsufe = Simplify[yfin / fe]

         "Freccia f=fe psi(alpha l) in sommita', con funzione non-lineare psi
          che esprime l'amplificazione di fe dovuta alla presenza del
          carico di compressione P";
         ffin = Simplify[yfin /. {x → l}]
         psi = Simplify[ffin / fe]
```

Out[45]= $\frac{1}{2\,\text{alpha}^4\,\text{EJ}}$ (Sec[alpha l] (alpha^2 x (-2 F + q (-2 l + x)) Cos[alpha l] + 2 (-q + alpha^2 W + (q - alpha^2 W) Cos[alpha x] + alpha (F + l q) Sin[alpha l] - alpha F Sin[alpha (l - x)] - alpha l q Sin[alpha (l - x)])))

Out[46]= (12 Sec[alpha l] (alpha^2 x (-2 F + q (-2 l + x)) Cos[alpha l] + 2 (-q + alpha^2 W + (q - alpha^2 W) Cos[alpha x] + alpha (F + l q) Sin[alpha l] - alpha F Sin[alpha (l - x)] - alpha l q Sin[alpha (l - x)]))) / (alpha^4 l^2 (8 F l + 3 l^2 q + 12 W))

Out[48]= $-\frac{1}{2\,\text{alpha}^4\,\text{EJ}}$ (2 alpha^2 F l - 2 q + alpha^2 l^2 q + 2 alpha^2 W + 2 (q - alpha^2 W) Sec[alpha l] - 2 alpha (F + l q) Tan[alpha l])

Out[49]= -(12 (2 alpha^2 F l - 2 q + alpha^2 l^2 q + 2 alpha^2 W + 2 (q - alpha^2 W) Sec[alpha l] - 2 alpha (F + l q) Tan[alpha l])) / (alpha^4 l^2 (8 F l + 3 l^2 q + 12 W))

```
In[50]:= "1) Caso particolare illustrativo:
           sola coppia W agente (F=0, q=0) con P";

         yfinW = Simplify[yfin /. {q → 0, F → 0}]
         yfinsufeW = Simplify[yfinsufe /. {q → 0, F → 0}]
         ffinW = Simplify[ffin /. {q → 0, F → 0}]
         psiW = Simplify[psi /. {q → 0, F → 0}]
```

Out[51]= $\frac{2\,W\,\mathrm{Sec}[\mathrm{alpha}\,l]\,\mathrm{Sin}\left[\frac{\mathrm{alpha}\,x}{2}\right]^2}{\mathrm{alpha}^2\,EJ}$

Out[52]= $-\frac{2\,(-1+\mathrm{Cos}[\mathrm{alpha}\,x])\,\mathrm{Sec}[\mathrm{alpha}\,l]}{\mathrm{alpha}^2\,l^2}$

Out[53]= $\frac{W\,(-1+\mathrm{Sec}[\mathrm{alpha}\,l])}{\mathrm{alpha}^2\,EJ}$

Out[54]= $\frac{2\,(-1+\mathrm{Sec}[\mathrm{alpha}\,l])}{\mathrm{alpha}^2\,l^2}$

```
In[55]:= "Caso W. Rappresentazione della linea elastica a meno del fattore fe
          per vari alpha l. Al crescere di P, la deformata si allontana
          sempre di piu' da quella elastica ottenibile per P=0, fino
          a divergere in sommita' per P -> Pcr";

         yfinsufeWplot = Simplify[yfinsufeW /. {x → xa l}];
         Plot[{yfinsufeWplot /. {alpha → 0.001 / l},
               yfinsufeWplot /. {alpha → 0.6 / l},
               yfinsufeWplot /. {alpha → 0.8 / l},
               yfinsufeWplot /. {alpha → 1.0 / l},
               yfinsufeWplot /. {alpha → 1.1 / l},
               yfinsufeWplot /. {alpha → 1.2 / l}}, {xa, 0, 1},
              PlotRange → {{0, 1}, {0, 2.5}},
              AxesLabel → {x / l, "yW/fe"}, AxesOrigin → {0, 0},
           PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]},
                        {Dashing[{ }], Hue[0.8], Thickness[0.005]},
                        {Dashing[{ }], Hue[0.7], Thickness[0.005]},
                        {Dashing[{ }], Hue[0.6], Thickness[0.005]},
                        {Dashing[{ }], Hue[0.5], Thickness[0.005]},
                        {Dashing[{ }], Hue[0.4], Thickness[0.005]}}]

         "Caso W. Rappresentazione del fattore di amplificazione di fe";

         Plot[{psiW /. alpha → alphal / l}, {alphal, 0, 1.2},
               PlotRange → {{0, 1.2001}, {0, 2.5}},
               AxesLabel → {alpha l, "psiW=f/fe"}, AxesOrigin → {0, 0},
               PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]}}]
```



Out[57]= - Graphics -



Out[59]= - Graphics -

In[60]:=
```
"2) Caso particolare illustrativo:
  sola forza F agente (W=0, q=0) con P";

yfinF = Simplify[yfin /. {q → 0, W → 0}]
yfinsufeF = Simplify[yfinsufe /. {q → 0, W → 0}]
ffinF = Simplify[ffin /. {q → 0, W → 0}]
psiF = Simplify[psi /. {q → 0, W → 0}]
Null
```

Out[61]= $\frac{F\,(-\text{alpha}\,x + \text{Sin}[\text{alpha}\,x] - (-1 + \text{Cos}[\text{alpha}\,x])\,\text{Tan}[\text{alpha}\,l])}{\text{alpha}^3\,EJ}$

Out[62]= $-\frac{3\,(\text{alpha}\,x + \text{Sec}[\text{alpha}\,l]\,\text{Sin}[\text{alpha}\,(l - x)] - \text{Tan}[\text{alpha}\,l])}{\text{alpha}^3\,l^3}$

Out[63]= $\frac{F\,(-\text{alpha}\,l + \text{Tan}[\text{alpha}\,l])}{\text{alpha}^3\,EJ}$

Out[64]= $\frac{-3\,\text{alpha}\,l + 3\,\text{Tan}[\text{alpha}\,l]}{\text{alpha}^3\,l^3}$

In[66]:=
```
"Caso F. Rappresentazione della linea elastica a meno del fattore fe
 per vari alpha l. Al crescere di P, la deformata si allontana
 sempre di piu' da quella elastica ottenibile per P=0, fino
 a divergere in sommita' per P -> Pcr";

yfinsufeFplot = Simplify[yfinsufeF /. {x → xa l}];
Plot[{yfinsufeFplot /. {alpha → 0.001 / l},
      yfinsufeFplot /. {alpha → 0.6 / l},
      yfinsufeFplot /. {alpha → 0.8 / l},
      yfinsufeFplot /. {alpha → 1.0 / l},
      yfinsufeFplot /. {alpha → 1.1 / l},
      yfinsufeFplot /. {alpha → 1.2 / l}}, {xa, 0, 1},
     PlotRange → {{0, 1}, {0, 2.5}},
     AxesLabel → {x / l, "yF/fe"}, AxesOrigin → {0, 0},
  PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]},
               {Dashing[{ }], Hue[0.8], Thickness[0.005]},
               {Dashing[{ }], Hue[0.7], Thickness[0.005]},
               {Dashing[{ }], Hue[0.6], Thickness[0.005]},
               {Dashing[{ }], Hue[0.5], Thickness[0.005]},
               {Dashing[{ }], Hue[0.4], Thickness[0.005]}}]

"Caso F. Rappresentazione del fattore di amplificazione di fe";

Plot[{psiF /. alpha → alphal / l}, {alphal, 0, 1.2},
      PlotRange → {{0, 1.2001}, {0, 2.5}},
      AxesLabel → {alpha l, "psiF=f/fe"}, AxesOrigin → {0, 0},
      PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]}}]
```



Out[68]= - Graphics -



Out[70]= - Graphics -

In[71]:=
```
"3) Caso particolare illustrativo:
  solo carico distribuito q agente (W=0, F=0) con P";

yfinq = Simplify[yfin /. {F → 0, W → 0}]
yfinsufeq = Simplify[yfinsufe /. {F → 0, W → 0}]
ffinq = Simplify[ffin /. {F → 0, W → 0}]
psiq = Simplify[psi /. {F → 0, W → 0}]
```

Out[72]=
$$\frac{1}{2\,\text{alpha}^4\,\text{EJ}}\left(q\left(2\,\text{Sec}[\text{alpha l}]\left(-1+\text{Cos}[\text{alpha x}]-\text{alpha l Sin}[\text{alpha}(\text{l}-\text{x})]\right)+\text{alpha}\left(\text{alpha x}(-2\,\text{l}+\text{x})+2\,\text{l Tan}[\text{alpha l}]\right)\right)\right)$$

Out[73]=
$$\frac{1}{\text{alpha}^4\,\text{l}^4}\left(8\,\text{Sec}[\text{alpha l}]\left(-1+\text{Cos}[\text{alpha x}]-\text{alpha l Sin}[\text{alpha}(\text{l}-\text{x})]\right)+4\,\text{alpha}\left(\text{alpha x}(-2\,\text{l}+\text{x})+2\,\text{l Tan}[\text{alpha l}]\right)\right)$$

Out[74]=
$$-\frac{q\left(-2+\text{alpha}^2\,\text{l}^2+2\,\text{Sec}[\text{alpha l}]-2\,\text{alpha l Tan}[\text{alpha l}]\right)}{2\,\text{alpha}^4\,\text{EJ}}$$

Out[75]=
$$\frac{8-4\,\text{alpha}^2\,\text{l}^2-8\,\text{Sec}[\text{alpha l}]+8\,\text{alpha l Tan}[\text{alpha l}]}{\text{alpha}^4\,\text{l}^4}$$

In[76]:=
```
"Caso q. Rappresentazione della linea elastica a meno del fattore fe
 per vari alpha l. Al crescere di P, la deformata si allontana
 sempre di piu' da quella elastica ottenibile per P=0, fino
 a divergere in sommita' per P -> Pcr";

yfinsufeqplot = Simplify[yfinsufeq /. {x → xa l}];
Plot[{yfinsufeqplot /. {alpha → 0.001 / l},
     yfinsufeqplot /. {alpha → 0.6 / l},
     yfinsufeqplot /. {alpha → 0.8 / l},
     yfinsufeqplot /. {alpha → 1.0 / l},
     yfinsufeqplot /. {alpha → 1.1 / l},
     yfinsufeqplot /. {alpha → 1.2 / l}}, {xa, 0, 1},
    PlotRange → {{0, 1}, {0, 2.5}},
    AxesLabel → {x / l, "yq/fe"}, AxesOrigin → {0, 0},
  PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]},
             {Dashing[{ }], Hue[0.8], Thickness[0.005]},
             {Dashing[{ }], Hue[0.7], Thickness[0.005]},
             {Dashing[{ }], Hue[0.6], Thickness[0.005]},
             {Dashing[{ }], Hue[0.5], Thickness[0.005]},
             {Dashing[{ }], Hue[0.4], Thickness[0.005]}}]

"Caso q. Rappresentazione del fattore di amplificazione di fe";

Plot[{psiq /. alpha → alphal / l}, {alphal, 0, 1.2},
     PlotRange → {{0, 1.2001}, {0, 2.5}},
     AxesLabel → {alpha l, "psiq=f/fe"}, AxesOrigin → {0, 0},
     PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]}}]
```



Out[78]= - Graphics -



Out[80]= - Graphics -

```
In[81]:= "Confronto tra i tre casi:
          I tre fattori di amplificazione psiW, psiF, psiq sono molto
          prossimi tra loro";

         plotpsi = Plot[{{psiW /. alpha → alphal / l},
                         {psiF /. alpha → alphal / l},
                         {psiq /. alpha → alphal / l}}, {alphal, 0, 1.2},
               PlotRange → {{0, 1.2001}, {0, 2.5}},
               AxesLabel → {alpha l, "psiW,psiF,psiq"}, AxesOrigin → {0, 0},
               PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]},
                            {Dashing[{ }], Hue[0.7], Thickness[0.005]},
                            {Dashing[{ }], Hue[0.5], Thickness[0.005]}}]
         plotpsi2 = Plot[{{psiW /. alpha → Sqrt[alphalq] / l},
                          {psiF /. alpha → Sqrt[alphalq] / l},
                          {psiq /. alpha → Sqrt[alphalq] / l}}, {alphalq, 0, 1.2},
               PlotRange → {{0, 1.2001}, {0, 2}},
               AxesLabel → {"P/EJ", "psiW,psiF,psiq"}, AxesOrigin → {0, 0},
               PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]},
                            {Dashing[{ }], Hue[0.7], Thickness[0.005]},
                            {Dashing[{ }], Hue[0.5], Thickness[0.005]}}]

         "Rappresentazione alternativa classica con assi invertiti
          con alpha l=Sqrt[P/EJ] e (alpha l)^2=P/EJ, legati a P,
          espressi in funzione dei tre fattori psi.
          Tale rappresentazione mostra la risposta carico/spostamento
          non lineare (per non-linearita' geometrica) con cosiddetto
          effetto P-Delta";

         Show[plotpsi /. x_Line :→ Map[Reverse, x, {2}],
              AxesLabel → {"psiW,psiF,psiq", "alpha l=Sqrt[P/EJ]"},
              PlotRange → {{1, 2.5}, {0, 1.2}},
              AxesOrigin → {1, 0},
               Ticks → {Automatic, Automatic},
              AspectRatio → Automatic];
         Show[plotpsi2 /. x_Line :→ Map[Reverse, x, {2}],
              AxesLabel → {"psiW,psiF,psiq", "(alpha l)^2=P/EJ"},
              PlotRange → {{1, 2.0}, {0, 1.201}},
              AxesOrigin → {1, 0},
               Ticks → {Automatic, Automatic},
              AspectRatio → Automatic];
```



Out[82]= - Graphics -



Out[83]= - Graphics -


alpha l=Sqrt[P/EJ]
1
0.8
0.6
0.4
0.2
1.2
1.4
1.6
1.8
2
2.2
2.4
psiW,psiF,psiq




```
In[87]:= "Calcolo del CARICO CRITICO EULERIANO dell'asta incastrata al piede
          senza carichi trasversali";
         ycr = y /. {W → 0, F → 0, q → 0};
         ycrp = D[ycr, x];

         "Imposizione delle condizioni al contorno al piede";
         ycrcc = Collect[Simplify[ycr /. {Solve[{ycr /. x → 0} == 0, B][[1, 1]],
               Solve[{ycrp /. x → 0} == 0, A][[1, 1]]}], f];
         "Imposizione della condizione ycr(l)=f";
         Simplify[f - {ycrcc /. x → l}][[1]]

Out[93]= f Cos[alpha l]
```

In[94]:=
```
"Soluzioni di Cos[alpha l]=0, k=1,2,3 ... ";
alphacr = (2 n - 1) Pi / 2 / l;
ycrn = ycrcc /. {alpha → alphacr}

"Rappresentazione delle prime tre deformate critiche";
Plot[{ycrn / f /. {n → 1, x → xa l},
      ycrn / f /. {n → 2, x → xa l},
      ycrn / f /. {n → 3, x → xa l}}, {xa, 0, 1},
      PlotRange → {{0, 1}, {0, 2}},
      AxesLabel → {x / l, "ycr/f"}, AxesOrigin → {0, 0},
      PlotStyle → {{Dashing[{ }], Hue[0.9], Thickness[0.005]},
              {Dashing[{ }], Hue[0.7], Thickness[0.005]},
              {Dashing[{ }], Hue[0.5], Thickness[0.005]}}]
```

Out[96]= $f\left(1-\mathrm{Cos}\left[\frac{(-1+2\,n)\,\pi\,x}{2\,l}\right]\right)$



Out[98]= - Graphics -